

✓ Ejercicio 6.31 (página 283)

✓ Ecuacion de la conica

✓ (%i1) `ec:x^2-6*x*y-7*y^2+10*x+2*y+9;`

[(%o1) $-7y^2 - 6xy + 2y + x^2 + 10x + 9$

✓ Forma matricial

✓ (%i2) `UXY:matrix([1],[x],[y]);`

[(%o2) $\begin{bmatrix} 1 \\ x \\ y \end{bmatrix}$

✓ (%i3) `UXYT:transpose(UXY);`

[(%o3) $\begin{bmatrix} 1 & x & y \end{bmatrix}$

✓ (%i4) `/*Matriz de la cónica*/`

`A:matrix([9,5,1],[5,1,-3],[1,-3,-7]);`

[(%o4) $\begin{bmatrix} 9 & 5 & 1 \\ 5 & 1 & -3 \\ 1 & -3 & -7 \end{bmatrix}$

✓ (%i5) `/* A es una matriz simétrica */`
`is(A = transpose(A));`

[(%o5) `true`

✓ (%i6) `UXYT.A.UXY;`

[(%o6) $(-7y - 3x + 1)y + y + x(-3y + x + 5) + 5x + 9$

✓ (%i7) `expand(%);`

[(%o7) $-7y^2 - 6xy + 2y + x^2 + 10x + 9$

✓ (%i8) `/* Comprobación */`
`ec - (UXYT.A.UXY);`

[(%o8) $-7y^2 - (-7y - 3x + 1)y - 6xy + y - x(-3y + x + 5) + x^2 + 5x$

✓ (%i9) `expand(%);`

[(%o9) 0

✓ Invariantes

✓ (%i10) `I3:determinant(A);`

[(%o10) 0

```
(%i11) Aoo:submatrix(1,A,1);
(%o11)  $\begin{bmatrix} 1 & -3 \\ -3 & -7 \end{bmatrix}$ 
```

```
(%i12) I2:determinant(Aoo);
(%o12) -16
```

```
(%i13) I1:A[2,2]+A[3,3];
(%o13) -6
```

Clasificación

```
(%i14) /*
Clasificación usando los invariantes:
Se trata de una cónica degenerada [det(A)=0]
Es una cónica con centro [Aoo distinto de cero]
Es del tipo de la hipérbola [Aoo < 0]
Son un par de rectas que se cortan.
*/
```

```
Ao1:submatrix(1,A,2); Ao2:submatrix(1,A,3);
(%o14)  $\begin{bmatrix} 5 & -3 \\ 1 & -7 \end{bmatrix}$ 
(%o15)  $\begin{bmatrix} 5 & 1 \\ 1 & -3 \end{bmatrix}$ 
```

Centro de la cónica = (C1, C2)

```
(%i16) C1:-determinant(Ao1)/determinant(Aoo);
(%o16) -2
```

```
(%i17) C2:determinant(Ao2)/determinant(Aoo);
(%o17) 1
```

Ecuación reducida usando invariantes

```
(%i18) /*Una vez clasificacada la matriz, sabemos
cómo es la forma reducida*/
Formareducida:matrix([0,0,0],[0,k1,0],[0,0,k2]);
(%o18)  $\begin{bmatrix} 0 & 0 & 0 \\ 0 & k1 & 0 \\ 0 & 0 & k2 \end{bmatrix}$ 
```

```
(%i19) /*los autovalores k1 y k2 son las soluciones del
polinomio caracteristico de Aoo
I1 = k1 + k2
I2 = k1*k2
*/
```

```
solve(m^2-I1*m+I2=0);
```

```
(%o19) [ m = - 8 , m = 2 ]
```

```
(%i20) /*
La forma reducida es

-8*x^2+2*y^2=0
*/

/*
Son dos rectas que se cortan
*/
```

```
factor(-8*x^2+2*y^2=0);
```

```
(%o20) 2(y - 2 x)(y + 2 x)=0
```

Representación grafica

```
(%i21) load(draw);
```

```
Loading C:/Users/ACER/maxima/binary/binary-gcl/share/draw/grcommon.o
```

```
Finished loading C:/Users/ACER/maxima/binary/binary-gcl/share/draw/grcommon.o
```

```
Loading C:/Users/ACER/maxima/binary/binary-gcl/share/draw/gnuplot.o
```

```
Finished loading C:/Users/ACER/maxima/binary/binary-gcl/share/draw/gnuplot.o
```

```
Loading C:/Users/ACER/maxima/binary/binary-gcl/share/draw/vtk.o
```

```
Finished loading C:/Users/ACER/maxima/binary/binary-gcl/share/draw/vtk.o
```

```
Loading C:/Users/ACER/maxima/binary/binary-gcl/share/draw/picture.o
```

```
Finished loading C:/Users/ACER/maxima/binary/binary-gcl/share/draw/picture.o
```


```
(%o21) C:/PROGRA~1/MAXIMA~1.2/share/maxima/5.31.2/share/draw/draw.lisp
```

```
(%i22) draw2d(
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[-4,2],
  yrange=[-2,4],
  color=blue,
  implicit(ec=0,x,-4,2,y,-2,4),
  head_length=0.1,
  vector([0,0],[0,1]),
  vector([0,0],[1,0]),
  color=red,
  vector([-2,1],[1/sqrt(10),3/sqrt(10)]),
  vector([-2,1],[-3/sqrt(10),1/sqrt(10)])
);
```

```
(%o22) [ gr2d( implicit , vector , vector , vector , vector ) ]
```

Gráfica

Figura 1: C:\Users\ACER\Desktop\AVIP\UNED\ALGEBRA\Mi material Tema 6\rectas



☑ C:\Users\ACER\Desktop\AVIP\UNED\ALGEBRA\Mi material Tema 6\rectas que se c

☑ Cálculo directo de la ecuación reducida

☑ Ecuación reducida mediante un giro y una translación

☑ (%i23) eigenvectors(Aoo);

(%o23) [[[-8, 2], [1, 1]], [[1, 3]], [[1, -1/3]]]

☑ (%i24) uniteigenvectors(Aoo);

(%o24) [[[-8, 2], [1, 1]], [[1/√10, 3/√10]], [[3/√10, -1/√10]]]

☑ (%i25) G:=matrix([1/sqrt(10), 3/sqrt(10)], [3/sqrt(10), -1/sqrt(10)]);

(%o25)
$$\begin{bmatrix} \frac{1}{\sqrt{10}} & \frac{3}{\sqrt{10}} \\ \frac{3}{\sqrt{10}} & -\frac{1}{\sqrt{10}} \end{bmatrix}$$

```
(%i26) determinant(G);
(%o26) -1
```

```
(%i27) /*Se cambia uno de los vectores columna para que la matriz de paso
ortogonal sea un giro, sin invertir la orientación*/

GP:matrix([1/sqrt(10), -3/sqrt(10)],[3/sqrt(10),1/sqrt(10)]);
(%o27) 
$$\begin{bmatrix} \frac{1}{\sqrt{10}} & -\frac{3}{\sqrt{10}} \\ \frac{3}{\sqrt{10}} & \frac{1}{\sqrt{10}} \end{bmatrix}$$

```

```
(%i28) XY:matrix([x],[y]); XYT:transpose(XY);
(%o28) 
$$\begin{bmatrix} x \\ y \end{bmatrix}$$

(%o29) 
$$\begin{bmatrix} x & y \end{bmatrix}$$

```

```
(%i30) GPT:transpose(GP);
(%o30) 
$$\begin{bmatrix} \frac{1}{\sqrt{10}} & \frac{3}{\sqrt{10}} \\ -\frac{3}{\sqrt{10}} & \frac{1}{\sqrt{10}} \end{bmatrix}$$

```

```
(%i31) /*Término cuadrático después del giro*/

ecg:XYT.GPT.Aoo.GP.XY;
(%o31) 
$$x \left( \frac{-3 \left( \frac{y}{\sqrt{10}} + \frac{3x}{\sqrt{10}} \right) - \frac{3y}{\sqrt{10}} + \frac{x}{\sqrt{10}}}{\sqrt{10}} + \frac{3 \left( -7 \left( \frac{y}{\sqrt{10}} + \frac{3x}{\sqrt{10}} \right) - 3 \left( \frac{x}{\sqrt{10}} - \frac{3y}{\sqrt{10}} \right) \right)}{\sqrt{10}} \right) + y$$


$$\left( \frac{-7 \left( \frac{y}{\sqrt{10}} + \frac{3x}{\sqrt{10}} \right) - 3 \left( \frac{x}{\sqrt{10}} - \frac{3y}{\sqrt{10}} \right)}{\sqrt{10}} - \frac{3 \left( -3 \left( \frac{y}{\sqrt{10}} + \frac{3x}{\sqrt{10}} \right) - \frac{3y}{\sqrt{10}} + \frac{x}{\sqrt{10}} \right)}{\sqrt{10}} \right)$$

```

```
(%i32) /*Término cuadrático después del giro simplificado*/

q:expand(ecg);
(%o32) 2 y^2 - 8 x^2
```

```
(%i33) /*Término lineal después del giro*/

l:matrix([10,2]).GP.XY;
(%o33) 
$$2 \left( \frac{y}{\sqrt{10}} + \frac{3x}{\sqrt{10}} \right) + 10 \left( \frac{x}{\sqrt{10}} - \frac{3y}{\sqrt{10}} \right)$$

```

```
(%i34) l:expand(l);
```

$$(\%o34) -3\sqrt{10}y + \frac{2y}{\sqrt{10}} + \sqrt{10}x + \frac{6x}{\sqrt{10}}$$

```
(%i35) l:ratsimp(l);
```

$$(\%o35) -\frac{14\sqrt{10}y - 8\sqrt{10}x}{5}$$

```
(%i36) /*La nueva ecuación de la conica después del giro*/
```

```
ec1:q+l+9;
```

$$(\%o36) 2y^2 - \frac{14\sqrt{10}y - 8\sqrt{10}x}{5} - 8x^2 + 9$$

```
(%i37) /* se efectúa una traslación (a,b)*/
```

```
e:subst([x=x+a, y=y+b], ec1);
```

$$(\%o37) 2(y+b)^2 - \frac{14\sqrt{10}(y+b) - 8\sqrt{10}(x+a)}{5} - 8(x+a)^2 + 9$$

```
(%i38) expand(e);
```

$$(\%o38) 2y^2 + 4by - \frac{14\sqrt{10}y}{5} - 8x^2 - 16ax + \frac{8\sqrt{10}x}{5} + 2b^2 - \frac{14\sqrt{10}b}{5} - 8a^2 + \frac{8\sqrt{10}a}{5} + 9$$

```
(%i39) /*Se busca el valor de b adecuado para anular los términos en y*/
```

```
solve([4*b-14*sqrt(10)/5],[b]);
```

$$(\%o39) [b = \frac{7}{\sqrt{10}}]$$

```
(%i40) /*Se busca el valor de a adecuado para anular los términos en x*/
```

```
solve([-16*a+8*sqrt(10)/5],[a]);
```

$$(\%o40) [a = \frac{1}{\sqrt{10}}]$$

```
(%i41) /* Se hace la traslación para los valores hallados*/
```

```
r:subst([a=1/sqrt(10), b=7/sqrt(10)], e);
```

$$(\%o41) 2\left(y + \frac{7}{\sqrt{10}}\right)^2 - \frac{14\sqrt{10}\left(y + \frac{7}{\sqrt{10}}\right) - 8\sqrt{10}\left(x + \frac{1}{\sqrt{10}}\right)}{5} - 8\left(x + \frac{1}{\sqrt{10}}\right)^2 + 9$$

```
(%i42) /* Ecuación reducida*/
```

```
ratsimp(r);
```

$$(\%o42) 2y^2 - 8x^2$$